

# SPOT


The Society of PET Owners and Trainers

EDITED BY PHYLLIS COLE



PET photo courtesy of Utter Chaos

*Commodore's PET is a factory assembled personal computer based on a 6502 microprocessor. The \$795 system includes a keyboard, cassette tape unit, built-in TV screen, some graphics, upper and lower case, and extended 8K BASIC, and 8K of user memory. Each bimonthly issue of People's Computers since the September-October 1977 issue has included an article on the PET. —PC*



---

## PET PEEVES

### CONLEY'S COMMENTS

On December 16, 1977, the Reverend David M Conley sent a letter to Commodore detailing a variety of problems that the Universal Life Church of the Pacific was experiencing with its PET, and attempts to set up an appointment to have it repaired. They were understandably reluctant to ship their system back to the factory for an unknown period of time, and they were hopeful of finding some other way of getting their PET to function more satisfactorily. Copies of the letter were sent to 5 publications and an attorney.

Six months later, a second letter was sent by the Reverend Conley to Commodore, pleading for some response. We sincerely hope that by now Commodore has dealt with the problems. One of the most discouraging items in the letters was the fact that Commodore had not, in a 6 month period, even sent out its instruction booklet.

### DOCUMENTATION DILEMMA

As indicated above, documentation — rather, the lack thereof — is a severe handicap for PET users. Many purchasers open their new PETs to find not a scrap of information, despite the fact that since last November various booklets, however slim, have in fact been available from the company. One reason given for the lack of documentation was that all copies of the previous booklet had been shipped; and, rather than print more of the old version, the new version would be shipped to these customers at a future date. This sort of explanation is hardly acceptable. As more and more publications try to fill the gap Commodore has left, some users will find that with time, effort and money the desired information can be accumulated. But the process is often slow and painful, and for those unsuspecting persons who buy a PET thinking they're getting an appliance, the process may be next to impossible to initiate. After all, if you've never used a computer, never read a computer magazine, and there's no computer store in your town, just where do you begin?

---

## COMMODORE CORNER

At last, the long-awaited guidelines for submission of software for publication by Commodore are available. When/if they reach the hands of those of you who've submitted software is another question. Personally, we've had, at best, erratic responses — usually none — in requesting information. One example: last November, *People's Computers* wrote to Commodore offering to compile a booklet of articles, programs, etc., that could be sent to new owners to provide some pointers about PET usage, since almost no documentation was available. We were prepared to do this at no expense to Commodore, as a way of getting some free advertising and also offering what we saw as a valuable

service. We received no response to the letter.

As indicated in the PET PEEVES section, others have been similarly frustrated in dealing with the company. One reason for this may be the very rapid turnover of personnel. We know of several employees who lasted only a few months; among them is Adrian Byram, the Software Manager whose name appears on the guides to software publication that we've seen.

We're discouraged with the company, though we still enjoy our PET. Perhaps the 'COMMODORE CORNER' can be summed up in a single word: confusing.

# PET POINTERS

## PRETTY PRINTING

The PET left-adjusts print functions, which is not very neat when printing columns of dollars and cents. The following few lines inserted in a program will print the value of the variable, A, right-adjusted with the decimal point in the 25th (or any) column.

```
120 A$=STR$(INT(A))
125 IF INT(ABS(A))<1 THEN A$="1"
130 PRINT TAB (25-LEN(A$));A
```

also

```
10 A=INT(100xA+.5)/100
```

will round off A to dollars and cents.

P.S. We love our PET. James Standley, Hastings College, Hastings, Nebraska 68901.

## SHARP TAPE DECK

In the Mar-Apr issue, page 55, TAPE TIPS suggests short tapes to find programs easily. The Sharp 1155 tape deck has an automatic program search system that looks for blank spaces between programs in either rewind or fast forward. By saving a computer program, then using the Sharp deck pause control to insert 10 secs of blank tape, the next program can be saved and any program can be found by counting the stops due to blank spaces. It takes only seconds to find the 9th or 10th program on a tape. Jack Clark, Oxton Hill, MD

## PET-HAM INTERFACE

Ron Lodewyck, co-author of Commodore's Basic BASIC tutorial tape (a nice job, from what we've heard) is offering a PET-Ham interface called the M-65. It allows you to send and receive Morse code as well as radio teletype code; it is also a Morse code trainer. The hardware plus software package is available in kit form for \$69.96, or \$99.95 assembled. For more information, write Microtronics, 5943 Pioneer Road, Hughson, CA 95326. Or telephone (209) 634-8888.

## MICROSIGNAL CATALOG

Another company offering PET peripherals and software is Microsignal, PO Box 161988, Sacramento, CA 95816 (see earlier issues of *People's Computers* for additional sources). Their catalog lists a voice input device, a sound generator, and a device that allows you to use a Teletype as a printer for the PET. Software for the most part exercises the peripherals offered by the company.




## RS-232 PRINTER ADAPTER

Connecticut microComputer announces the first in a line of peripheral adapters for the Commodore PET. The PET ADAPTER model 1200 drives an RS-232 printer from the PET IEEE-488 bus. The PET ADA 1200 allows the PET owner to obtain hard copy program listings, and to type letters, manuscripts, mailing labels, tables of data, etc., using a standard RS-232 printer.

The PET ADA model 1200 is available assembled and tested, without power supplies, case, or RS-232 connector for \$98.50 (plus \$5.00 for shipping and handling) or complete for \$169 (plus \$5.00 for shipping and handling). Specify baud rate when ordering. (300 baud is supplied unless otherwise requested.) Contact: Connecticut microComputer, 150 Pocono Rd., Brookfield, Ct; (203) 775-9659

## LIGHT PEN

A self-contained light pen which plugs directly into the Commodore PET 2001 user port has been announced by the 3G Company. This light pen makes it possible to bypass the PET's keyboard and interact directly with the information displayed on the CRT screen.



The light pen adds versatility to most graphics programs and is valuable as a teaching aid for young children. It also adds unique capabilities for application programs aimed at the non-computer oriented person. The light pen is complete and ready to plug into the PET. A sampled program and programming instructions comes with the pen. The entire package sells for \$24.95 and is available from 3G Company at Rt.3, Box 28A, Gaston, Oregon 97119; (503)985-7176

## PET LISTING CONVENTIONS

PET Program listings in *People's Computers* employ the following conventions to represent characters that are difficult to print on a standard printer: Whenever square brackets appear in the listing, neither the brackets nor the text they enclose should be typed literally. Instead, the text between the brackets should be translated to keystrokes. For example, [CLR] means type the CLR key, [3 DOWN] means [DOWN, DOWN, DOWN] i.e., press the first CRSR key three times.

## PET BLOOPERS

Woops! Several readers have informed us that the PLOT program by Philip Gash which we published in last issue's (V7, No.1) SPOT section will not run correctly as published. Indeed, lines 560 and 620 should be corrected as follows: (the underscore marks the corrected error):

```
560 X=D
620 INPUT "MAX VALUE OF
Y(X)"; Y4
```

## POSTSCRIPT TO VIDEO MIXER

Last issue, in my article on the 'PET Video Mixer,' I mistakenly wrote that the three video signals to be mixed into a composite video output came from the PET IEEE 488-bus. This is NOT the case; the signals actually come from pinouts on the PARALLEL USER PORT. For references to the 'IEEE 488-bus' in the article and the schematic diagram, one should substitute PARALLEL USER PORT. Also note that there is a drawing error in one of the wires going to a pin on IC<sub>1</sub>. The wire that goes from the 20k ohm trim pot should have been drawn connected to pin 7 of IC<sub>1</sub>, not pin 6. Indeed, the circuit will not work properly if pin 6 instead of pin 7 is wired to +5 volts. Also omitted from the schematic is indication that the *positive* side of the 100 microfarad cap is connected to the 8-9V power supply through a 75 ohm resistor.

One fellow PET user who used this schematic to wire up his own video mixer pointed out to me recently that this circuit does not work with all monitors. This circuit, as is, will work only with monitors (or modified TV circuits) that require a POSITIVE video voltage in. If you need a negative video sync voltage for your particular monitor, then you could substitute a 7404 or 7416 IC for IC<sub>3</sub>. 7407 is an inverter IC and is designed to produce a positive-going video signal. Besides determining whether you need a positive or negative-going out-put signal for the mixer to your monitor, you will also need to experiment with what voltage level works best with your particular monitor. If you have further questions or want more information about the Video Mixer, please contact me.

Randall Julin, 15 Poncetta Dr #322,  
Daly City, CA 94015 (415) 469-  
1157 (Day) (415)922-6946 (Home)

# HORSES

BY ANDY STADLER

*This horse-race game was written by 12-year old Andy Stadler, of Albany, California. I met Andy at a local PET User's Group meeting; thanks for letting us share your program with People's Computers' readers, Andy.*

*HORSES lets from 1 to 9 players bet on one of 10 horses. Odds start at 10 to 1. If a horse wins, its odds go down by 1; if a horse loses, its odds increase by 1 (see lines 1140 and 1150).*

*As published, the 'track' for each horse*

*prints across the screen from left to right. If you prefer a faster track (heh heh, she said, in fair imitation of ye honorable editor Kahn) 500 PRINT "[RVS, 40 SPACE, OFF]";*


*One friend who tried out the game insisted upon whistling the traditional 'they're off and running' music once the horses were at the gate. To allow time for this, you may wish to insert a pause at the appropriate time by including the line:*

*605 FOR X=1 to 500: NEXT X*

*in your program.*

— PC





## HORSES

```

10 POKE 59468,14
20 REM ***HORSES***BY ANDY STADLER***
30 FOR X=1 TO 10: A(X)=10: NEXT X
40 PRINT "[CLR]"; SPC(10); "HORSES"
50 PRINT "[3 DOWN]HAVE ALL PLAYERS PLAYED BEFORE?";
60 GET A$: IF A$="" THEN 60
70 IF A$<>"Y" AND A$<>"N" THEN 60
80 PRINT CHR$(ASC(A$)+128)
90 IF A$="Y" THEN 230
100 PRINT "[UP] THIS IS THE GAME OF HORSES. IN IT,"
110 PRINT "[DOWN]YOU BET ON ONE OF THE 10 HORSES. EACH"
120 PRINT "[DOWN]STARTS WITH ODDS OF 10:1. IF A"
130 PRINT "[DOWN]HORSE WINS, ITS ODDS GO DOWN BY 1."
140 PRINT "[DOWN]IF A HORSE LOSES A RACE, ITS"
150 PRINT "[DOWN]ODDS GO UP BY 1. YOU START WITH 200"
160 PRINT "[DOWN]DOLLARS. EACH TIME YOU BET, THE "
170 PRINT "[DOWN]ONLY LIMIT IS YOUR BANKROLL!"
180 PRINT "[2 DOWN]DO YOU UNDERSTAND?"
190 GET A$: IF A$="" THEN 190
200 IF A$<>"N" THEN 220
210 PRINT "n": PRINT "THEN FIND SOMEONE WHO DOES,"
    YOU $8\#": GOTO 1120
220 PRINT A$
230 PRINT "[CLR]"; SPC(10); "HORSES"
240 PRINT "[3 DOWN]HOW MANY PLAYERS (1-9)? ";
250 GET B: IF B=0 THEN 250
260 PRINT B
270 PRINT "[DOWN]": FOR X=1 TO B
280 PRINT "PLAYER #";X;"NAME: "; P$(X)=""
290 PRINT "[ε,LEFT]";
300 GET XX$: IF XX$="" THEN 300
310 IF ASC(XX$)=13 THEN PRINT "[2 SPACE]": GOTO 340
320 LET P$(X)=P$(X)+XX$: PRINT XX$;
330 GOTO 290
340 LET B(X)=200: NEXT X
350 PRINT "[CLR]"; SPC(10); "HORSES"
360 PRINT "[2 DOWN]THE HORSES ARE: [DOWN]"
370 RESTORE
380 FOR X=1 TO 10
390 READ A$
400 PRINT "#"; X; ": "; A$;TAB(25);"ODDS:";A(X);"TO 1"
410 NEXT X: PRINT "[DOWN]";
420 FOR X=1 TO B: IF B(X)<=0 THEN 470
430 PRINT P$(X);", # HORSE, BET": INPUT C,D
440 IF C<1 OR C>10 OR D<0 OR D>B(X) THEN 460
450 LET Z(X)=C: LET Y(X)=D: GOTO 470
460 PRINT "[UP]"; : GOTO 430
470 NEXT X
480 PRINT "[CLR, 2 DOWN]";
490 FOR X=1 TO 11
500 FOR Q=1 TO 40
510 PRINT "[RVS, SPACE, OFF]";
520 NEXT Q

```

```

530 PRINT "[DOWN]";
540 NEXT X
550 PRINT "[HOME]"
560 PRINT "[2 DOWN]";
570 FOR X=1 TO 10
580 LET C(X)=1
590 PRINT X; "[2 DOWN, 3 LEFT]";
600 NEXT X
610 PRINT "[HOME]";
620 REM***THEY'RE OFF!***
630 PRINT "THEY'RE OFF! [4 SPACE, HOME]";
640 LET W=INT(10*RND(1)+1): IF C(5)=5 THEN
    PRINT "[HOME, 13 SPACE, [HOME]";
650 FOR X=1 TO (W*2)+1
660 PRINT "[DOWN]";
670 NEXT X
680 LET C(W)=C(W)+1
690 PRINT SPC(C(W)-1); W; "[HOME]";
700 IF C(W)<39 THEN 640
710 PRINT "THE WINNER: ";:RESTORE:FOR X=1 TO W
720 READ A$: NEXT X: PRINT A$;" (NO. "; W;)"":
    FOR X=1 TO 5000:NEXT X
730 REM**COMPUTE WINS AND LOSSES**
740 PRINT "[CLR]";:FOR X=1 TO B: IF B(X)<=0 THEN 810
750 PRINT "[DOWN]";P$(X);",YOU ";
760 IF Z(X)=W THEN PRINT "WON";
770 IF Z(X)<>W THEN PRINT "LOST";
780 IF Z(X)=W THEN PRINT A(W)*Y(X);
790 IF Z(X)<>W THEN PRINT Y(X);
800 PRINT "DOLLARS."
810 NEXT X
820 FOR X=1 TO B:IF B(X)<=0 THEN 860
830 IF Z(X)=W THEN LET B(X)=B(X)+(A(W)*Y(X))
840 IF Z(X)<>W THEN LET B(X)=B(X)-Y(X)
850 IF B(X)<=0 THEN PRINT "[DOWN]"; P$(X);
    " YOU BUSTED!"
860 NEXT X
870 FOR X=1 TO 5000:NEXT X
880 PRINT "[CLR, 6 SPACE] THE STANDINGS: [DOWN]":
    LET Q=0
890 FOR X=1 TO B: IF B(X)<=0 THEN 920
900 PRINT P$(X); TAB(10); "$";B(X)
910 GOTO 930
920 Q=Q+1:GOTO 940
930 Z9=X
940 NEXT X: IF Q=B-1 AND B>1 GOTO 1110
950 IF Q=B THEN PRINT "YOU'RE ALL DEAD!": END
960 PRINT "[DOWN] ANOTHER ROUND? ";
970 GET A$: IF A$="" GOTO 970
980 IF A$<>"Y" AND A$<>"N" GOTO 970
999 IF A$="Y" GOTO 1130
1000 PRINT A$: GOTO 1170
1010 DATA "FIREBALL"
1020 DATA "SEATTLE SLOW"
1030 DATA "ZAPPING ZING"
1040 DATA "FRED'S FOLLY"
1050 DATA "WORRYSOME WART"
1060 DATA "HELL'S ANGEL"
1070 DATA "HEAVEN'S DEVIL"
1080 DATA "TELLY'S SON"
1090 DATA "WHAT A HOPE!"
1100 DATA "KEEP WISHING!"
1110 PRINT "[3 DOWN]"; P$(Z9);",YOU WON!!! WITH $";
    B(Z9); "!!!"
1115 PRINT "JOLLY GOOD SHOW!!!"
1120 GOTO 1170
1130 FOR X=1 TO 10
1140 IF X=W THEN A(X)=A(X)-1
1150 IF X<W THEN A(X)=A(X)+1
1160 NEXT X: GOTO 350
1170 PRINT "[3 DOWN] COME AGAIN SOON!":END

```

# M U J E B L

BY M.C. HOFHEINZ

*M.C. Hofheinz' 'kids' programs are definitely for kids of all ages. In a previous issue, we published several short graphics programs; here's a word-oriented one. Note that in lines 2000 and 2002, various numbers of blanks are inserted to keep words from being printed partly on one line and partly on another. The program requires more than one screenfull to display all the combinations for 5- and 6-letter words. You may wish to study some of the combinations before they scroll out of sight: just press the STOP key. To continue, type the word CONT then press the return key.* — PC

```

20  REM: PROGRAM NAME "JUMBLE"
30  REM: WRITTEN BY M.C. HOFHEINZ
40  REM: STOCKTON CALIF, MAY 1,1978
50  REM: ASSIGN A Z(Y) VALUE TO EACH LETTER
100 PRINT "TYPE ANY WORD OF 6 OR FEWER LETTERS"
150 INPUT N$
200 FOR Y=1 TO LEN (N$)
300 Z$(Y) = MID$(N$,Y,1)
400 NEXT Y
500 REM: START NESTED LOOPS TO INTERCHANGE LETTERS
600 FOR A=1 TO LEN(N$)
700 FOR B=1 TO LEN(N$)
710 REM: LEN(N$)TESTS FOR WORDS SHORTER THAN 6 LETTERS
715 IF LEN(N$)<3 THEN 1000
800 FOR C=1 TO LEN (N$)
805 IF LEN(N$)<4 THEN 1000
900 FOR D=1 TO LEN(N$)
905 IF LEN(N$)<5 THEN 1000
910 FOR E=1 TO LEN(N$)
915 IF LEN(N$)<6 THEN 1000
920 FOR G=1 TO LEN(N$)
990 REM: SORT SO THAT EACH LETTER IS PRINTED ONLY ONCE
1000 IF A=B THEN 2300
1025 IF LEN(N$)<3 THEN 2000
1100 IF A=C OR B=C THEN 2200
1225 IF LEN(N$)<4 THEN 2000
1250 IF A=D or B=D or C=D THEN 2100
1450 IF LEN(N$)<5 THEN 2000
1500 IF A=E OR B=E OR C=E OR D=E THEN 2070
1800 IF LEN(N$)<6 THEN 2000
1820 IF A=G OR B=G OR C=G OR D=G OR E=G THEN 2050
2000 PRINT Z$(A); Z$(B); Z$(C); Z$(D); Z$(E); Z$(G); " "; :IF LEN(N$)=5
      THEN PRINT " ";
2002 IF LEN(N$)=6 THEN PRINT " ";
2005 IF LEN(N$)<3 THEN 2300
2010 IF LEN(N$)<4 THEN 2200
2020 IF LEN(N$)<5 THEN 2100
2030 IF LEN(N$)<6 THEN 2070
2050 NEXT G
2070 NEXT E
2100 NEXT D
2200 NEXT C
2300 NEXT B
2400 NEXT A

```

Another program for the kids. This one takes any word (or number) up to 6 characters and prints out every possible combination of those characters. The program was originally designed to help my kids solve the JUMBLE scrambled word puzzle which appears in the *San Francisco Chronicle*, and other papers.

A peculiarity of the PET makes it necessary to use A, B, C, D, E, and G (not F) since 'F OR' merges into FOR and produces a SYNTAX ERROR (see line 1820). This is due to the fact that FOR is an instruction word. □

